

# **ABAP OBJECTS – BASICS**


## **GUIDED TUTORIAL**

Exercises / Solutions by Peter McNulty, Horst Keller / SAP

# Introduction

This exercise is a tutorial or guided tour through the fundamental language elements of ABAP Objects and the usage of the respective ABAP Workbench tools. The tutorial is designed for developers who have had little or no experience with ABAP and ABAP Objects until now.

The following class diagram shows the scope of the exercise.



(After finishing the exercise, you should be able to display this diagram for your own classes).

## Exercise 1, Classes and Objects

### (Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_A)


#### Create a vehicle class

Create a class ZCL\_VEHICLE\_XX (where XX is your group number).

- The class should have the protected instance attributes SPEED and MAX\_SPEED for its speed and maximum speed, and the public methods SPEED\_UP, STOP, and SHOW. Furthermore there should be a private attribute that contains a running number for an object ID.
- SPEED\_UP should have an IMPORTING parameter STEP. The method should increase the speed by STEP, but not allow it to exceed the maximum speed.
- STOP should reset the speed to zero.
- WRITE should display a message containing the speed and the maximum speed.


#### Solution

1. Logon to the system and open the Object Navigator of the ABAP Workbench (Transaction SE80, enter /nSE80 in the command field of the system task bar).
2. Select **Local Objects** in order to work in a test package that is not transported to other systems.




Hit **Enter**.

3. Right Click the name of the local package and navigate to the creation of a class (where XX is your group number).




4. Fill the pop-up as follows ZCL\_VEHICLE\_XX (where XX is your group number) and select **Save**.




(Don't forget to uncheck **Final**)

5. Acknowledge the following window without changes (select either **Save** or **Local Object**).





The same holds for all other development objects during this exercise.

6. Now you enter the Class Builder



Here you can edit the class either in **Form-Based** mode (default) or in **Source Code-Based** mode. Use the respective button to toggle between the modes.

7. Switch to **Source Code-Based** mode , switch to **Change** mode  and replace the existing template with the following code (where XX is your group number).




```
CLASS zcl_vehicle_xx DEFINITION PUBLIC CREATE PUBLIC.
  PUBLIC SECTION.
    METHODS constructor.
    METHODS speed_up
    IMPORTING
```


```


        step TYPE i.
    METHODS stop.
    METHODS show.
PROTECTED SECTION.
    DATA: speed      TYPE i,
           max_speed TYPE i VALUE 50.
PRIVATE SECTION.
    DATA id TYPE i .
    CLASS-DATA object_count TYPE i.
ENDCLASS.

CLASS zcl_vehicle_xx IMPLEMENTATION.
METHOD constructor.
    object_count = object_count + 1.
    id = object_count.
ENDMETHOD.
METHOD show.
    DATA msg TYPE string.
    msg = `Vehicle `      && |{ id }| &&
           ` , Speed = `  && |{ speed }| &&
           ` , Max-Speed = ` && |{ max_speed }|.
    MESSAGE msg TYPE 'I'.
ENDMETHOD.
METHOD speed_up.
    speed = speed + step.
    IF speed > max_speed.
        speed = max_speed.
    ENDIF.
ENDMETHOD.
METHOD stop.
    speed = 0.
ENDMETHOD.
ENDCLASS.


```

Check,  Save  and Activate  the class (acknowledge all entries of the activation pop-up ).

8. Switch back to **Form-Based** mode  and play around in that mode by double clicking the class components.



Tip: Select Signature when displaying the implementation of a method.




### Create an application class

Create a class ZCL\_APPLICATION\_XX (where XX is your group number).


- This class should use your vehicle class.
- It should have simply one static method START that creates and uses objects of ZCL\_VEHICLE\_XX.

### Solution

1. Create the class in the object navigator, where you can directly right click **Classes** now



2. Fill the pop-up as follows (where XX is your group number).






3. Implement the class as follows:


```
CLASS zcl_application_xx DEFINITION PUBLIC ABSTRACT FINAL CREATE PUBLIC.
  PUBLIC SECTION.
    CLASS-METHODS start.
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.
```

```
CLASS ZCL_APPLICATION_XX IMPLEMENTATION.
  METHOD start.
    DATA vehicle      TYPE REF TO zcl_vehicle_xx.
    DATA vehicle_tab LIKE TABLE OF vehicle.
    DATA tabindex TYPE i.
    DO 10 TIMES.
      CREATE OBJECT vehicle.
      APPEND vehicle TO vehicle_tab.
    ENDDO.
    LOOP AT vehicle_tab INTO vehicle.
      tabindex = sy-tabix * 10.
      vehicle->speed_up( tabindex ).
      vehicle->show( ).
    ENDO.
  ENDMETHOD.
ENDCLASS.
```






Check,  Save  and Activate  the class (acknowledge all entries of the activation pop-up ).


4. Select  (F8) in the Class Builder and execute method START.



Now you can see the Ids and speed of the objects created.



5. To examine further, navigate to the source code of method START  and create a breakpoint at an appropriate position. To add a breakpoint at a certain line, double-click in the left margin on the line that you want a breakpoint. A little  should appear in the margin on that line. Similarly, this is the same way you remove a breakpoint.



Test the method again and play around with the ABAP Debugger.

## Exercise 2, Inheritance

### (Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_B)


#### Create a truck subclass

Create a subclass ZCL\_TRUCK\_XX (where XX is your group number).


- The class should have an instance constructor that sets the maximal speed to 100
- It should redefine the method SHOW to produce some specific output.

#### Solution

1. Create the class in the object navigator as before, but select **Create inheritance** (  ) this time in order to enter a **Superclass**.




Save  and Activate .



2. Enter the truck's own constructor in the **Form-based** mode (type it or select **Create constructor**  ), double click it and implement it as follows:

```
METHOD constructor.  
    super->constructor( ).  
    max_speed = 100.  
ENDMETHOD.
```

Save  and Activate .

3. Select method SHOW in the **Form-based** mode and redefine it by selecting . Replace the implementation as follows:

```
METHOD show.
  DATA msg TYPE string.
  msg = `Truck `      &&
        ` Speed = `  && |{ speed }| &&
        ` Max-Speed = ` && |{ max_speed }|.
  MESSAGE msg TYPE 'I'.
ENDMETHOD.
```

4. **Save**  and **Activate** .

Check out the syntax in the **Source code-based** mode.

### Create a ship subclass


Create a subclass ZCL\_SHIP\_XX (where XX is your group number).

- The class should have an instance constructor that sets the maximal speed to 30 and that has an import parameter to set an additional read-only attribute to the ship's name.
- It should redefine the method SHOW to produce some specific output (including the ship's name).

### Solution

1. Create the subclass ZCL\_SHIP\_XX in the object navigator as you did before for the truck class (where XX is your group number)..


2. Add a new public instance attribute NAME of type STRING either in **Form-based** mode or in **Source code-based** mode.



Attribute	Level	Visibility	Read-Only	Typing	Associated Type	Description	Initial value
<ZCL_VEHICLE_99>			<input type="checkbox"/>				
SPEED	Instance Attribute	Protected	<input type="checkbox"/>	Type	I		
MAX_SPEED	Instance Attribute	Protected	<input type="checkbox"/>	Type	I		50
NAME	Instance Attribute	Public	<input type="checkbox"/>	Type	STRING		

3. Insert the ship's own constructor as you did for the truck class.

4. Create an importing parameter for the constructor either by selecting **Parameter** in **Form-based** mode



Param	Pass	O	Typing M	Associated Type	Default value	Descrip
NAME	<input type="checkbox"/>	<input type="checkbox"/>	Type	STRING		

or by adding it in **Source code-based** mode:



```
METHODS constructor
IMPORTING
  name TYPE string.
```

5. Implement the constructor as follows:

```
METHOD CONSTRUCTOR.  
  super->constructor( ).  
  max_speed = 30.  
  me->name = name.  
ENDMETHOD.
```

6. Redefine method SHOW as follows:

```
METHOD show.  
  DATA msg TYPE string.  
  msg = me->name &&  
    ` , Speed = ` && |{ speed }| &&  
    ` , Max-Speed = ` && |{ max_speed }|.  
  MESSAGE msg TYPE 'I'.  
ENDMETHOD.
```

7. **Save**  and **Activate** .

Check out the syntax in the **Source code-based** mode.

### Adjust the application class

The code of the START method should demonstrate the usage of the subclasses now.

- Declare extra reference variables TRUCK and SHIP for the new classes.
- You can delete the code that creates objects for VEHICLE. Instead, create one instance of each of your new subclasses and place the corresponding reference into VEHICLE\_TAB.
- Call the method SPEED\_UP for both classes using the respective subclass reference, and SHOW using a superclass reference.

### Solution

1. Replace the code of method START of ZCL\_APPLICATION\_XX with the following (where XX is your group number):

```
METHOD start.  
  DATA: vehicle TYPE REF TO zcl_vehicle_xx,  
        vehicle_tab LIKE TABLE OF vehicle,  
        truck TYPE REF TO zcl_truck_xx,  
        ship TYPE REF TO zcl_ship_xx.  
  CREATE OBJECT: truck,  
                ship EXPORTING name = 'Titanic'.  
  APPEND: truck TO vehicle_tab,  
         ship TO vehicle_tab.  
  truck->speed_up( 30 ).  
  ship->speed_up( 10 ).  
  LOOP AT vehicle_tab INTO vehicle.  
    vehicle->show( ).  
  ENDLIST.  
ENDMETHOD.
```

Note the polymorphic method call `vehicle->show( )`.

2. Execute method START from the Class Builder again.

## Exercise 3, Interfaces



### (Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_C)

#### Create a status interface


- Create an interface ZIF\_STATUS\_XX (where XX is your group number).
- The interface should have one instance method SHOW.

#### Solution

1. Create the interface in the object navigator as follows:



2. Define one Method without parameters:



Save  and Activate .


### Implement the interface in the superclass

Implement ZIF\_STATUS\_XX in ZCL\_VEHICLE\_XX (where XX is your group number).

- Copy the implementation of the class method SHOW to the interface method SHOW.
- Delete class method SHOW
- Create an alias name SHOW for the interface method in order to keep the subclasses valid.

### Solution

1. Open ZCL\_VEHICLE\_XX and enter the interface either in **Form-based** mode:




or in **Source code-based** mode:

```
PUBLIC SECTION.
  INTERFACES zif_status_xx.
```

2. Implement the interface method with the code from SHOW:

```
METHOD zif_status_xx~show.
  DATA msg TYPE string.
  msg = `Vehicle `      && |{ id }| &&
        ` , Speed = `  && |{ speed }| &&
        ` , Max-Speed = ` && |{ max_speed }|.
  MESSAGE msg TYPE 'I'.
ENDMETHOD.
```

3. Delete (🗑️) method SHOW.
4. Create a public alias SHOW:



5. Save (💾) and Activate (🔧) ..
6. Check ZCL\_TRUCK\_XX and ZCL\_SHIP\_XX (where XX is your group number). They redefine the superclass interface method SHOW via the alias SHOW now and there should be no errors.

## Create a new helicopter class that also implements the interface

Create a new class ZCL\_HELICOPTER\_XX (where XX is your group number) that is not part of the ZCL\_VEHICLE\_XX inheritance tree but that also implements STATUS.

### Solution

1. Create class ZCL\_HELICOPTER\_XX (where XX is your group number) as follows:

```
CLASS zcl_helicopter_xx DEFINITION PUBLIC FINAL CREATE PUBLIC .
  PUBLIC SECTION.
    INTERFACES zif_status_xx.
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.

CLASS zcl_helicopter_xx IMPLEMENTATION.
  METHOD zif_status_xx~show.
    DATA msg TYPE string.
    msg = `Helicopter, idle`.
    MESSAGE msg TYPE 'I'.
  ENDMETHOD.
ENDCLASS.
```

2. Save  and Activate .

### Adjust the application class

The code of the START method should demonstrate the usage of the interface now.

- Declare a reference variable HELI for the class ZCL\_HELICOPTER\_XX following (where XX is your group number) and create a corresponding object.
- Replace the reference variable VEHICLE and the table VEHICLE\_TAB with an interface reference STATUS and an internal table STATUS\_TAB.
- Insert the reference variables for truck, ship and helicopter into the table STATUS\_TAB.

### Solution

1. Replace the code of method START of ZCL\_APPLICATION\_XX with the following (where XX is your group number):

```
METHOD start.
  DATA: status      TYPE REF TO zif_status_xx,
        status_tab  LIKE TABLE OF status,
        truck       TYPE REF TO zcl_truck_xx,
        ship        TYPE REF TO zcl_ship_xx,
        heli        TYPE REF TO zcl_helicopter_xx.
  CREATE OBJECT: truck,
                ship EXPORTING name = 'Titanic',
                heli.
  APPEND: truck TO status_tab,
         ship TO status_tab,
         heli TO status_tab.
  truck->speed_up( 30 ).
  ship->speed_up( 10 ).
  LOOP AT status_tab INTO status.
    status->show( ).
  ENDLIST.
ENDMETHOD.
```

Note the polymorphic method call `status->show( )`.

2. Execute method START from the Class Builder again.



## Exercise 4, Events

### (Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_D)


#### Define an event in the ship class

Objects of ZCL\_SHIP\_XX (where XX is your group number) should raise an event if the speed becomes higher than the maximal speed.

- Define an instance event DAMAGED in ZCL\_SHIP\_XX
- Raise the event in method SPEED\_UP.

#### Solution

1. Open ZCL\_SHIP\_XX (where XX is your group number) in the Class Builder and define an instance event DAMAGED either in **Form-based** mode:



Event	Type	Visibility
DAMAGED	Instance Event	Public

or in **Source code-based** mode:

```
PUBLIC SECTION.  
...  
EVENTS damaged.
```

2. Redefine (  ) method SPEED\_UP and implement it as follows:

```
METHOD SPEED_UP.  
  speed = speed + step.  
  IF speed > max_speed.  
    max_speed = 0.  
    CALL METHOD stop.  
    RAISE EVENT damaged.  
  ENDIF.  
ENDMETHOD.
```

3. **Save**  and **Activate** .


#### Define an event handler method in the helicopter class


The helicopter class should be able to handle the event DAMAGED of the ship class.

- Define a public method RECEIVE as an event handler for event DAMAGED of ZCL\_SHIP\_XX in ZCL\_HELICOPTER\_XX (where XX is your group number).
- Implement the event handler in a way that simply sends a message that uses the default event parameter SENDER to address the name of the damaged ship .


## Solution

1. Open ZCL\_HELICOPTER\_XX (where XX is your group number) in the class builder.

In **Form-based** mode, enter a new public method RECEIVE and select **Detail view** (  ) where you can define the method as an **Event handler**:



and define the predefined importing parameter SENDER:





Parameter	Type	Pa	O	Typing M	Associated Type	Default value	Description
SENDER	Importin	<input type="checkbox"/>	<input type="checkbox"/>				>SENDER<

In **Source code-based** mode it is enough to type:

```
METHODS receive
  FOR EVENT damaged OF zcl_ship_xx.
    IMPORTING sender.
```

2. Implement method RECEIVE as follows:

```
METHOD receive.  
  DATA msg TYPE string.  
  msg = `Helicopter received call from ` && sender->name.  
  MESSAGE msg TYPE 'I'.  
ENDMETHOD.
```

3. Save  and Activate .

### Adjust the application class

The code of the START method should register the instances of the helicopter class for the events of the ship class.

- Use statement SET HANDLER to register the event handler.
- Call method SPEED\_UP of the ship object in order to raise the event.

### Solution

1. Replace the code of method START of ZCL\_APPLICATION\_XX with the following (where XX is your group number):

```
METHOD start.  
  DATA: status      TYPE REF TO zif_status_xx,  
        status_tab  LIKE TABLE OF status,  
        truck       TYPE REF TO zcl_truck_xx,  
        ship        TYPE REF TO zcl_ship_xx,  
        heli        TYPE REF TO zcl_helicopter_xx.  
  CREATE OBJECT: truck,  
                ship EXPORTING name = 'Titanic',  
                heli.  
  APPEND: truck TO status_tab,  
        ship TO status_tab,  
        heli TO status_tab.  
  SET HANDLER heli->receive FOR ALL INSTANCES.  
  truck->speed_up( 30 ).  
  ship->speed_up( 10 ).  
  LOOP AT status_tab INTO status.  
    status->show( ).  
  ENDLOOP.  
  DO 5 TIMES.  
    ship->speed_up( 10 ).  
  ENDDO.  
ENDMETHOD.
```

2. Execute method START from the Class Builder again.

# Exercise 5, Exceptions

## (Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_E)


### Define an exception class


Objects of ZCL\_VEHICLE\_XX should raise an exception ZCX\_OVERHEATING\_XX (where XX is your group number) if the speed becomes higher than the maximal speed.



- Define an exception ZCX\_OVERHEATING\_XX (where XX is your group number)
- Raise the exception in method SPEED\_UP.

### Solution

1. Open ZCL\_VEHICLE\_XX (where XX is your group number) in the Class Builder, place the cursor on method SPEED\_UP in **Form-based** mode and select the button **Exception**.
2. Mark **Exception Classes** and enter ZCX\_OVERHEATING\_XX (where XX is your group number).



3. **Save**  and select **Yes** and **Save** on the following pop-ups with respectively.



4. Double click the name of exception class, **Activate** it and return to the vehicle class.
5. Change the implementation if method SPEED\_UP as follows (where XX is your group number)::

```
METHOD speed_up.
    speed = speed + step.
    IF speed > max_speed.
        RAISE EXCEPTION TYPE zcx_overheating_xx.
    ENDIF.
ENDMETHOD.
```

6. Save  and Activate .




### Adjust the application class

The code of the START method must handle the exception now.

- Use a TRY control structure to handle the exception ZCX\_OVERHEATING\_XX (where XX is your group number)
- Call method SPEED\_UP of the truck object in order to raise the exception.

### Solution

1. Check the syntax of method START of ZCL\_APPLICATION\_XX (where XX is your group number). You should get the following warnings:

Description	Row	Type
Class ZCL_APPLICATION_XX,Method START	20	
The exception ZCX_OVERHEATING_XX is neither caught nor is it declared in the RAISING clause of"START".		
Class ZCL_APPLICATION_XX,Method START	15	
The exception ZCX_OVERHEATING_XX is neither caught nor is it declared in the RAISING clause of"START".		
Class ZCL_APPLICATION_XX,Method START	14	
The exception ZCX_OVERHEATING_XX is neither caught nor is it declared in the RAISING clause of"START".		

2. Adjust the implementation of method START as follows (where XX is your group number):

```
METHOD start.
    DATA: status      TYPE REF TO zif_status_xx,
           status_tab LIKE TABLE OF status,
           truck      TYPE REF TO zcl_truck_xx,
           ship       TYPE REF TO zcl_ship_xx,
           heli       TYPE REF TO zcl_helicopter_xx.
    CREATE OBJECT: truck,
                  ship EXPORTING name = 'Titanic',
                  heli.
    APPEND: truck TO status_tab,
           ship TO status_tab,
           heli TO status_tab.
    SET HANDLER heli->receive FOR ALL INSTANCES.
    TRY.
        truck->speed_up( 30 ).
        ship->speed_up( 10 ).
        LOOP AT status_tab INTO status.
            status->show( ).
        ENDLOOP.
    DO 5 TIMES.
```

```

        ship->speed_up( 10 ).
    ENDDO.
    DO 5 TIMES.
        truck->speed_up( 30 ).
        truck->show( ).
    ENDDO.
    CATCH zcx_overheating_xx.
        MESSAGE 'Truck overheated' TYPE 'I'.
    EXIT.
ENDTRY.
ENDMETHOD.

```

No syntax warnings should occur any more.

- Execute method START from the Class Builder again.

## Exercise 6, Unit Tests

(Solution in Package Z\_ABAP\_OBJECTS\_INTRODUCTION\_F)


### Define a test class with a test method for the vehicle class

Class ZCL\_VEHICLE\_XX (where XX is your group number) should contain a test class that tests the SPEED\_UP method completely.

- Declare and implement a local test class TEST\_VEHICLE in the vehicle class.

### Solution

- Open ZCL\_VEHICLE\_XX (where XX is your group number) in the Class Builder and select the following:



- Enter the following code there (where XX is your group number):

```

CLASS test_vehicle DEFINITION DEFERRED.
CLASS zcl_vehicle_xx DEFINITION LOCAL FRIENDS test_vehicle.

```

The second statement is the important one. It declares the test class as friend of the global class in order to have access of the vehicle's private components. The first statement is necessary for the ABAP Compiler. (select F1 to learn more).

3. Navigate back to the Class Builder's main screen and select **Local Test Classes** in the above menu, answer the pop-up with **Yes**, and implement your test class as follows (where XX is your group number):



```
CLASS test_vehicle DEFINITION FOR TESTING
    RISK LEVEL HARMLESS
    DURATION SHORT.

PRIVATE SECTION.
    DATA vehicle TYPE REF TO zcl_vehicle_xx.
    METHODS: test_speed_up FOR TESTING,
             setup,
             teardown.

ENDCLASS.

CLASS test_vehicle IMPLEMENTATION.
METHOD setup.
    CREATE OBJECT vehicle.
ENDMETHOD.
METHOD test_speed_up.
    TRY.
        vehicle->speed_up( 50 ).
        cl_abap_unit_assert=>assert_equals(
            EXPORTING
                exp = 50
                act = vehicle->speed
                msg = 'Speed not as expexted'
                level = if_aunit_constants=>critical ).
    CATCH zcx_overheating_xx.
        cl_abap_unit_assert=>fail(
            EXPORTING
                msg = 'No exception expected'
                level = if_aunit_constants=>critical ).
    ENDTRY.
    TRY.
        vehicle->speed_up( 1000 ).
        cl_abap_unit_assert=>fail(
            EXPORTING
                msg = 'Exception expected'
                level = if_aunit_constants=>critical ).
    CATCH zcx_overheating_xx.
    ENDTRY.
ENDMETHOD.
METHOD teardown.
    CLEAR vehicle.
ENDMETHOD.
ENDCLASS.
```

**Save**

4. Navigate back to the Class Builder's main screen, **Save**  and **Activate** .
5. Carry out the test method by

**Class -> Unit Test** in the Class Builder menu

or

**Test -> Unit Test** from the context menu of the class in the Repository Browser

The result should be:


✔ Processed successfully: 1 programs, 1 test classes, 1 test methods

### Use the ABAP Unit Browser


Examine the possibilities of the ABAP Unit Browser embedded in the Object Navigator

#### Solution

1. Select **Utilities -> Settings** in the Object Navigator and additionally select the **ABAP Unit Test Browser** under **Workbench (General)**:




2. Select the ABAP Unit Browser:




3. Select Class Pool, enter ZCL\_VEHICLE\_XX (where XX is your group number), and carry out the test with measuring the test coverage:





4. After successful execution you can navigate through the results to view the test coverage:



Result Edit System Help

**ABAP Unit: Result Display**

Program Level Coverage by Statements

Cov...	Proc. Cov.	Stmnt Cov.	Tcode/Program	Obj.	Object Name
▲	50,00	58,33	ZCL_VEHICLE_XX=====CP	CLAS	ZCL_VEHICLE_XX

Procedure Level Coverage by Statements

Cov.	Stmnt Cov.	Type	Name	Class of the Processing Block	Statements	Invocation
■	100,00	METH	CONSTRUCTOR	ZCL_VEHICLE_XX	3	1
■	100,00	METH	SPEED_UP	ZCL_VEHICLE_XX	4	2
●	0,00	METH	STOP	ZCL_VEHICLE_XX	2	0
●	0,00	METH	ZIF_STATUS_XX~SHOW	ZCL_VEHICLE_XX	3	0

M30 (2) 800 iwdfrm3022 INS

Statement Coverage Edit Goto System Help

**ABAP Unit: Statement Coverage**

Program ZCL\_VEHICLE\_XX

```

1  METHOD speed up.
2      speed = speed + step.
3      IF speed > max_speed.
4          RAISE EXCEPTION TYPE zcx_overheating_xx.
5      ENDIF.
6  ENDMETHOD.

```

M30 (2) 800 iwdfrm3022 INS

# Exercise 7, Service Enablement

## (Solution in Package ZABAP\_OO\_SERVICEENABLEMENT)

### Expose Method As Web Service (Inside-Out Approach)


In this exercise,

- You will create a RFC-enabled Function Module to invoke the IF\_DEMO\_CR\_CAR\_RENTL\_SERVICE~MAKE\_RESERVATION method from Class CL\_DEMO\_CR\_CAR\_RENTAL\_SERVICE
- Copy existing RFC-enabled Function Module into your Function Group
- Expose the RFC-enabled Function Modules in the Function Group as a Web Service
- Configure/Test the web service using SOAManager and Web Service Navigator
- Optionally, debug using an external breakpoint

Class Interface		CL_DEMO_CR_CAR_RENTAL_SERVICE		Implemented / Active	
Properties Interfaces Friends Attributes <b>Methods</b> Events Types Aliases					
Parameter Exception <input type="checkbox"/> Filter					
Method	Level	Visibility	Method type	Description	
IF_DEMO_CR_CAR_RENTL_SERVICE~CREATE_CUSTOMER	Instance Method	Public		Create Customer	
IF_DEMO_CR_CAR_RENTL_SERVICE~GET_CARS_BY_CATEGORY	Instance Method	Public		Get Rental Cars by Category	
IF_DEMO_CR_CAR_RENTL_SERVICE~GET_CUSTOMER_BY_ID	Instance Method	Public		Get Customer by ID	
IF_DEMO_CR_CAR_RENTL_SERVICE~GET_CUSTOMERS_BY_NAME	Instance Method	Public		Get customers by name	
IF_DEMO_CR_CAR_RENTL_SERVICE~GET_RESERVATIONS_BY_CUST_ID	Instance Method	Public		Get Reservation by Customer ID	
<b>IF_DEMO_CR_CAR_RENTL_SERVICE~MAKE_RESERVATION</b>	Instance Method	Public		Create Reservation	
GET_SERVICE	Static Method	Public		Get Singleton for Service	
CLASS_CONSTRUCTOR	Static Method	Public		Static Constructor	

### Solution


1. Logon to the system and open the Object Navigator of the ABAP Workbench (Transaction SE80, enter /nSE80 in the command field of the system task bar).
2. Select **Local Objects** in order to work in a test package that is not transported to other systems.




Hit Enter.

3. Create a Function Group

- Right Click the name of the local package and navigate to the creation of a Function Group.
- Enter **ZFG\_XX** (where XX is your group number) and short text. Click **Save**.




- Acknowledge the following window without changes (select either Save or Local Object).




4. Create the function module.


- Expand the Function Groups Folder. Right click on the function group, **ZFG\_XX**, and choose **Create->Function Module**



- Enter **zCustomerReserveCar\_XX** (where XX is your group number) and short text. Click **Save**.



- You can disregard the following informational message whenever it may appear



5. Define the function module interface by entering its parameters

- Provide the import parameters for the function module under **Import** tab

Parameter Name	Typing	Associated Type	Pass Value
CUSTOMER_ID	TYPE	DEMO_CR_CUSTOMER_ID	<input checked="" type="checkbox"/>
CATEGORY	TYPE	DEMO_CR_CATEGORY	<input checked="" type="checkbox"/>
STARTDATE	TYPE	DEMO_CR_DATE_FROM	<input checked="" type="checkbox"/>
ENDDATE	TYPE	DEMO_CR_DATE_TO	<input checked="" type="checkbox"/>

Function module ZCUSTOMERRESERVECAR\_XX Inactive (Revised)

Attributes **Import** Export Changing Tables Exceptions Source code

Parameter Name	Typing	Associated Type	Default value	Optional	Pass Value	Short text
CUSTOMER_ID	TYPE	DEMO_CR_CUSTOMER_ID		<input type="checkbox"/>	<input checked="" type="checkbox"/>	Customer ID
CATEGORY	TYPE	DEMO_CR_CATEGORY		<input type="checkbox"/>	<input checked="" type="checkbox"/>	Car Category
STARTDATE	TYPE	DEMO_CR_DATE_FROM		<input type="checkbox"/>	<input checked="" type="checkbox"/>	Start Date
ENDDATE	TYPE	DEMO_CR_DATE_TO		<input type="checkbox"/>	<input checked="" type="checkbox"/>	End Date

- Provide the export parameters for the function module under **Export** tab

Parameter Name	Typing	Associated Type	Pass Value
RESERVATIONS	TYPE	DEMO_CR_RESERVATIONS_TT	<input checked="" type="checkbox"/>
RETURN	TYPE	BAPI_MSG	<input checked="" type="checkbox"/>

Function module ZCUSTOMERRESERVECAR\_XX Active

Attributes Import **Export** Changing Tables Exceptions Source code

Parameter Name	Typing	Associated Type	Pass Val...	Short text
RESERVATIONS	TYPE	DEMO_CR_RESERVATIONS_TT	<input checked="" type="checkbox"/>	Table Type for Reservation
RETURN	TYPE	BAPI_MSG	<input checked="" type="checkbox"/>	Message Text

- Please mark “Remote-enabled Module” radio button under **Attributes** tab.

Function module ZCUSTOMERRESERVECAR\_XX Inactive (Revised)

Attributes **Import** Export Changing Tables Exceptions Source code

Classification

Function Group ZFG\_XX Function Group for ABAP OO - Group XX

Short Text Reserve Car for Customer

Processing Type

Normal Function Module

Remote-Enabled Module  BasXML supported

Update Module

Start immedi.

Immediate Start, No Restart

Start Delayed

Coll.run

General Data

Person Responsible ABAP\_00\_99

Last Changed By ABAP\_00\_99

Changed on 13.09.2009

Package \$TMP

Program Name SAPLZFG\_XX

INCLUDE Name LZFG\_XXU01


Original Language EN

Not released

Edit Lock

Global

6. Complete the source code for your function module under Source Code tab.

<p>Function module: ZCUSTOMERRESERVECAR_XX Inactive</p> <p>Attributes Import Export Changing Tables Exceptions Source code</p> <pre> 1 FUNCTION zcustomerreservecar_xx . 2 ----- 3 *""Local Interface: 4 *" IMPORTING 5 *" VALUE(CUSTOMER_ID) TYPE DEMO_CR_CUSTOMER_ID 6 *" VALUE(CATEGORY) TYPE DEMO_CR_CATEGORY 7 *" VALUE(STARTDATE) TYPE DEMO_CR_DATE_FROM 8 *" VALUE(ENDDATE) TYPE DEMO_CR_DATE_TO 9 *" EXPORTING 10 *" VALUE(RESERVATIONS) TYPE DEMO_CR_RESERVATIONS_TT 11 *" VALUE(RETURN) TYPE BAPI_MSG 12 ----- 13 Insert code here 14 15 ENDFUNCTION.</pre>	<p>You can toggle between Display and Change mode using the  icon.</p>
---	--

- Add the ABAP source code

```
DATA: lr_service TYPE REF TO if_demo_cr_car_rentl_service,
      lo_exception TYPE REF TO cx_root,
      l_customer TYPE demo_cr_scustomer,
      l_reservation TYPE demo_cr_sreservation.
```

```
lr_service = cl_demo_cr_car_rental_service=>get_service( ).
```

TRY.

```
lr_service->make_reservation(
  EXPORTING
    i_customer_id = customer_id
    i_category     = category
    i_date_from   = startdate
    i_date_to     = enddate ).
```

```
CATCH cx_demo_cr_no_customer cx_demo_cr_lock cx_demo_cr_reservation INTO lo_exception.
```

```
return = lo_exception->get_text( ).
EXIT.
```

```
CATCH cx_root INTO lo_exception.
return = lo_exception->get_text( ).
EXIT.
```

ENDTRY.

```
l_customer = lr_service->get_customer_by_id( customer_id ).
```


```
* Method returns ALL reservations for this customer
reservations = lr_service->get_reservations_by_cust_id( customer_id ).
* Delete the records not matching the requested start and end dates
DELETE reservations WHERE date_from <> startdate OR date_to <> enddate.
* We still may have the situation where the same reservation request was booked multiple times
```

```
CASE lines( reservations ).
  WHEN 0.
    return = `Unable to confirm reservation - Contact Help Desk`.
  WHEN 1.
    READ TABLE reservations INTO l_reservation INDEX 1.
    return = `Reservation ` && l_reservation-
reservation_id && ` Booked for ` && l_customer-name.
  WHEN OTHERS.
    sort reservations by reservation_id ascending.
    return = `Multiple reservation exist for ` && l_customer-
name && ` in the selected time period`.
ENDCASE.
```


- Check,  Save  and Activate  the Function Module

7. Copy 3 additional Function Modules from a different package into your Function Group


- Navigate to Package - ZABAP\_OO\_SERVICEENABLEMENT click Display 




- Expand the Function Groups folder to display the Function Modules



- Right-click on the ZCUSTOMERCREATE\_XX Function Module and Select Copy




- Enter the Following
  - To Function Module **ZCUSTOMERCREATE\_99** (replace 99 with your group number)
  - Function Group **ZFG\_99** (replace 99 with your group number)
  - Select **Copy**




- You can ignore any pop-up informational messages
- Repeat the same steps above to copy 2 more Function Modules remembering to replace 99 with your group number

Copy From	Copy To
ZCUSTOMERFINDBYNAME_XX	ZCUSTOMERFINDBYNAME_99
ZCUSTOMERGETRESERVATIONS_XX	ZCUSTOMERGETRESERVATIONS_99

- Select Local Objects from the dropdown box




- Expand the Function Groups and Function Modules to Display your copied objects






- Double-click on **ZCUSTOMERCREATE\_XX** (where XX is your group number) and then click **Activate**




- Select the entire worklist and click continue




8. Using the Web Service Wizard, generate an Enterprise Service Definition for the function group ZFG\_XX. (where XX is your Group Number).


- Right mouse click on the function group, **ZFG\_XX**, and choose Create->Other Objects->Enterprise Service




- On the Provide Service Definition Details step, input the name of your service definition – **ZCustomerCarRental\_XX**(where XX is your Group Number). You can also enter a short description (Kurzbeschreibung) and set the Endpoint Type to Function Group. Click **Continue**




- On the Choose Endpoint step, you have to specify the name of the function group which will serve as the implementation for this service definition. If you want to use Name Mapping (underscores are removed and changed to camel case) you can **check the Mapping of names** option.




- On the *Choose Operations* step, you can select which operations you wish to expose. Click Continue




- On the *Configure Service* step, set the **PRF\_DT\_IF\_SEC\_LOW** profile in order to set the lowest security level for this service definition. Be sure to check *Deploy Service*. If you forget to check this box, you can complete this step later, manually from transaction code **SOAMANAGER** by creating an Endpoint.



- On the *Enter Package/Request* step, please check *Local Object* to save the generated Service Definition as local/private.




- On the final step choose *Complete*.




- You can optionally explore the Service Definition that was generated by the Wizard.


**Display Service Definition ZCustomerCarRental\_99**




- To start the SOA Manager, use the transaction code `SOAMANAGER`. ( enter `/n SOAManager` in the command field of the system task bar). SOA Manager is used to complete the configuration of service providers and consumer proxies in a local ABAP system.




- The transaction **SOAMANAGER** should launch a Web Dynpro ABAP based application in your Internet Browser. Choose the *Service Administration* tab and then click on *Single Service Administration*.




- In the Web Service Administration screen that comes up, you can search for your Service Definition (Hint use wildcard `Z*XX` where `XX` is your group number)




- Select the row in the Search Results for your Service Definition and then click the *Apply Selection* button. The bottom half of the screen will now show the details for the selected Service Definition.




- There are lots of changes that can be made to the Service Definition from this screen. You can see a summary of all the settings from the *Details* tab. You can alter the published classification settings from the *Classification* tab. From the *Configurations* tab, you see the Endpoints for this service. If you had forgotten to Deploy Service during the wizard, you would now have to create this Endpoint manually. The settings for the Endpoint were generated for us based upon the security profile we choose during the wizard.



- Go Back to Design Time Details




- Click on the **Overview** tab
- Click on **Open Web Service Navigator** for the selected binding




- A Separate Browser Window is started. If you are asked to login, enter the credentials Tester / abcd1234




- The url of wsdl is displayed. Click Next




- If prompted for a login to download the wsdl – Please enter your userid and pwd for M30 and click **OK**




- Select the **ZCUSTOMERCREATE\_XX** operation (where XX is your group number) and Click **Next**



- Enter your Group Number for the Customer Name and Click **Next**




- The system displays the input parameters and the result of the test.




- Click on the **Operation** link







- Select the **ZCUSTOMERFINDBYNAME\_XX** operation (where XX is your group number) and Click **Next**



- Enter your Group Number for the Customer Name Click **Next**




- The system displays the input parameters and the result of the test.




- **Important** Please make note of your Customer ID (we will use this later).


- Click on the **Operation** link




- Select the **ZCUSTOMERRESERVECAR\_XX** operation (where XX is your group number) and Click **Next**




- Enter a Category A and the CustomerID returned from the **ZCUSTOMERFINDBYNAME\_XX** operation. Please be sure to enter the Start and End dates using the **YYYY-MM-DD** ISO date format, Click **Next**





- The system displays the input parameters and the result of the test.



- Optional – Set an External Breakpoint in Function Module ZCUSTOMERRESERVECAR\_XX (where XX is your group number) by right clicking in the left margin and **Set External Breakpoint** after which you can debug from the WS Navigator into the ABAP system



- You can step thru the code using . The ABAP debugger has many powerful features and learning how to use it is essential for Java programmers accessing ABAP Business Logic.



© 2009 by SAP AG.

All rights reserved. SAP, R/3, SAP NetWeaver, Duet, PartnerEdge, ByDesign, SAP Business ByDesign, and other SAP products and services mentioned herein as well as their respective logos are trademarks or registered trademarks of SAP AG in Germany and other countries.

Business Objects and the Business Objects logo, BusinessObjects, Crystal Reports, Crystal Decisions, Web Intelligence, Xcelsius, and other Business Objects products and services mentioned herein as well as their respective logos are trademarks or registered trademarks of Business Objects S.A. in the United States and in other countries. Business Objects is an SAP company.

All other product and service names mentioned are the trademarks of their respective companies. Data contained in this document serves informational purposes only. National product specifications may vary.

These materials are subject to change without notice. These materials are provided by SAP AG and its affiliated companies ("SAP Group") for informational purposes only, without representation or warranty of any kind, and SAP Group shall not be liable for errors or omissions with respect to the materials. The only warranties for SAP Group products and services are those that are set forth in the express warranty statements accompanying such products and services, if any. Nothing herein should be construed as constituting an additional warranty.